NAG Library Routine Document

G05RJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05RJF generates pseudorandom uniform variates with joint distribution of a Frank Archimedean copula.

2 Specification

```
SUBROUTINE G05RJF (N, M, THETA, SORDER, STATE, X, LDX, SDX, IFAIL)
INTEGER            N, M, SORDER, STATE(*), LDX, SDX, IFAIL
REAL (KIND=nag_wp) THETA, X(LDX,SDX)
```

3 Description

Generates n pseudorandom uniform m-variates whose joint distribution is the Frank Archimedean copula C_θ, given by

$$C_\theta = -\frac{1}{\theta}\ln\left[1+\frac{\left(e^{-\theta u_1}-1\right)\left(e^{-\theta u_2}-1\right)\cdots\left(e^{-\theta u_m}-1\right)}{\left(e^{-\theta}-1\right)^{m-1}}\right], \quad \begin{cases}\theta\in(0,\infty),\\ u_j\in(0,1], & j=1,\ldots m;\end{cases}$$

with the special case:

$C_\infty = \min(u_1,u_2,\ldots,u_m)$, the Fréchet–Hoeffding upper bound.

The generation method uses mixture of powers.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05RJF.

4 References

Marshall A W and Olkin I (1988) Families of multivariate distributions *Journal of the American Statistical Association* **83** 403

Nelsen R B (2006) *An Introduction to Copulas* (2nd Edition) Springer Series in Statistics

5 Arguments

1: N – INTEGER *Input*

On entry: n, the number of pseudorandom uniform variates to generate.

Constraint: N ≥ 0.

2: M – INTEGER *Input*

On entry: m, the number of dimensions.

Constraint: M ≥ 2.

3: THETA – REAL (KIND=nag_wp) *Input*

On entry: θ, the copula parameter.

Constraint: THETA $\geq 1.0\times10^{-6}$.

4: SORDER – INTEGER *Input*

On entry: determines the storage order of variates; the (i,j)th variate is stored in $\mathrm{X}(i,j)$ if SORDER $= 1$, and $\mathrm{X}(j,i)$ if SORDER $= 2$, for $i = 1,2,\ldots,n$ and $j = 1,2,\ldots,m$.

Constraint: SORDER $= 1$ or 2.

5: STATE$(*)$ – INTEGER array *Communication Array*

Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

6: X(LDX, SDX) – REAL (KIND=nag_wp) array *Output*

On exit: the pseudorandom uniform variates with joint distribution described by C_θ, with $\mathrm{X}(i,j)$ holding the ith value for the jth dimension if SORDER $= 1$ and the jth value for the ith dimension of SORDER $= 2$.

7: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G05RJF is called.

Constraints:

if SORDER $= 1$, LDX $\geq$ N;
if SORDER $= 2$, LDX $\geq$ M.

8: SDX – INTEGER *Input*

On entry: the second dimension of the array X as declared in the (sub)program from which G05RJF is called.

Constraints:

if SORDER $= 1$, SDX $\geq$ M;
if SORDER $= 2$, SDX $\geq$ N.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, corrupt STATE argument.

IFAIL $= 2$

On entry, invalid THETA: THETA $= \langle value \rangle$.
Constraint: THETA $\geq 1.0 \times 10^{-6}$.

IFAIL $= 3$

On entry, N $= \langle value \rangle$.
Constraint: N ≥ 0.

IFAIL $= 4$

On entry, M $= \langle value \rangle$.
Constraint: M ≥ 2.

IFAIL $= 5$

On entry, invalid SORDER.
Constraint: SORDER $= 1$ or 2.

IFAIL $= 7$

On entry, LDX must be at least $\langle value \rangle$: LDX $= \langle value \rangle$.

IFAIL $= 8$

On entry, SDX must be at least $\langle value \rangle$: SDX $= \langle value \rangle$.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05RJF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

In practice, the need for numerical stability restricts the range of θ such that:

the routine requires $\theta \geq 1.0 \times 10^{-6}$;

if $\theta > -\ln\epsilon$, the routine returns pseudorandom uniform variates with C_∞ joint distribution;

where ϵ is the ***machine precision*** returned by X02AJF.

10 Example

This example generates thirteen four-dimensional variates for copula $C_{4.0}$.

10.1 Program Text

```
    Program g05rjfe

!     G05RJF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g05kff, g05rjf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: lseed = 1, nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: theta
      Integer                          :: genid, ifail, ldx, lstate, m, n,     &
                                          sdx, sorder, subid
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: x(:,:)
      Integer                          :: seed(lseed)
      Integer, Allocatable             :: state(:)
!     .. Executable Statements ..
      Write (nout,*) 'G05RJF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!     Initial call to initializer to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!     Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Read in sample size, number of dimensions and order
      Read (nin,*) n, m, sorder

      If (sorder==1) Then
!       X(N,M)
        ldx = n
        sdx = m
      Else
!       X(M,N)
        ldx = m
        sdx = n
      End If
```

```
      Allocate (x(ldx,sdx))

!     Read in parameter
      Read (nin,*) theta

!     Generate variates
      ifail = 0
      Call g05rjf(n,m,theta,sorder,state,x,ldx,sdx,ifail)

!     Display the variates
      If (sorder==1) Then
!       X(N,M)
        ifail = 0
        Call x04caf('General',' ',n,m,x,ldx,                                  &
          'Uniform variates with copula joint distribution',ifail)
      Else
!       X(M,N)
        ifail = 0
        Call x04caf('General',' ',m,n,x,ldx,                                  &
          'Uniform variates with copula joint distribution',ifail)
      End If

    End Program g05rjfe
```

10.2 Program Data

```
G05RJF Example Program Data
1  1  1762543          :: GENID,SUBID,SEED(1)
13  4  1               :: N,M,SORDER
4.0                    :: THETA
```

10.3 Program Results

```
 G05RJF Example Program Results

 Uniform variates with copula joint distribution
           1       2       3       4
  1   0.5679  0.1977  0.8682  0.2664
  2   0.0965  0.3532  0.9773  0.3102
  3   0.5526  0.2562  0.6341  0.6267
  4   0.8036  0.4747  0.7310  0.5515
  5   0.2043  0.9797  0.3628  0.4968
  6   0.4777  0.8146  0.3922  0.4005
  7   0.4162  0.5002  0.5074  0.2008
  8   0.3703  0.0971  0.0527  0.0278
  9   0.4354  0.4880  0.4096  0.4259
 10   0.2693  0.1169  0.0639  0.1555
 11   0.0127  0.3080  0.2352  0.4659
 12   0.0730  0.3239  0.2020  0.0568
 13   0.2369  0.0817  0.3118  0.4370
```